Maple Tutorial II

CS 310 — Fall 1999

This handout has two purposes. The first is learn how to use files with Maple. The second is to see how to solve a complicated problem using Maple.

1 Using Files

In this section we’ll look at inputting a file containing Maple commands into Maple. Later in the course, as we explore more and more complicated problems with Maple, instead of having you tediously type long equations, we’ll just provide a file containing the Maple commands for you to load up and execute as described here.

1. Open Notepad by going to Start -> Programs -> Accessories -> Notepad

2. Put the following lines into the file:

restart;
f1 := (a + b)^3;
f2 := expand(f1);

3. Now save the file on your CAE drive as sample.txt.

4. Start up Maple. We want to load into Maple the file that you created, so that it will execute the commands contained in it. To do this, simply type: read "f:/sample.txt"; from within Maple. The Maple commands in the file are executed and printed to the screen.

5. You can continue to add commands to sample.txt, then read the file again to execute them in Maple. It is best to start your files with restart; to wipe out all previous variables.

Try adding these commands to your file, then execute them in Maple.

db := diff(f2, b);
db := factor(db);

6. You can add comments to your Maple command files by beginning a line with the “#” character. For example,

# Compute the derivative with respect to a
da := diff(f2, a);

2 Output

We now describe how to print and save your work in Maple.

1. The best way to print your Maple session is to go to “File” and select ‘Print”.

You’ll get a new subwindow. From the subwindow, you can either output to a PostScript file, or directly print your Maple session. When a file is output to a PostScript file, it can be viewed using GS View 2.6, which is located in

Start -> CAE Applications -> Utilities.

2. To save your Maple session, so you can load it again later, use the “Save As” option under the “File” menu. To load it again later, use “Open” and select the file you saved. Files saved with this method should have a “.mws” extension.

3. Note that the above instructions save/print both commands and graphs that you may have created in your worksheet. Also, you can remove unwanted commands from your worksheet by moving to the line you want to remove, holding down Ctrl, and pressing the Del key.

3 Solving a Problem with Maple

Description of the Problem

The mechanism we’ll look at is a pendulum driving an arm, as shown in Figure 1. The pendulum arm is that denoted by c. As it swings, it moves arm d. The right end of arm d is constrained to move horizontally only. The vertical distance between points A and B is fixed at 1 meter.

The motion of the pendulum is given by the equation T (t) = a cos(bt) (1) where T is a function of the time t, a is the maximum angle in the swing, and b is the frequency.

What we want to find is a formula giving the maximum and minimum extent of the position x, for particular values of a, c and d, and also the velocity of x when T = 0 for general a, b, c and d.

Determining the Maximum and Minimum Values of x

The geometric relations between the angles T and P , and the lengths c, d and x are: c cos(T ) + d sin(P ) = 1 d cos(P ) − c sin(T ) = x

Figure 1: A diagram of the pendulum

We first set eq1 and eq2 to be these two equations, then form a list EQ containing both equations. Before you start, you may want to use the restart; command to clear old variables.

eq1 := c*cos(T) + d*sin(P) = 1;

eq2 := d*cos(P) - c*sin(T) = x;

EQ := {eq1, eq2};

We now want to find the values for x when the pendulum has its maximum and minimum values, which we will take to be T = π/4 and −π/4 (i.e., ±45^o). We will use the particular case when c = 0.8 and d = 0.7.

So, we substitute in EQ (which is both equations) the values T = π/4, c = 0.8, d = 0.7 to find the maximum value. (In Maple, the symbol for π is Pi). Then, we solve the equations for the remaining unknowns, x and P . Similarly, to find the minimum value we change in the above T = π/4 to T = −π/4 and resolve the equations.

EQ0 := subs(T=Pi/4, c=0.8, d=0.7, EQ);

sol := solve(EQ0, {x,P});

What are the maximum and minimum values of x if T has maximum and minimum

±π/3? The answers are, to 2 decimal places, x_{max} = 1.05 and x_{min} = −0.33.

Determine the values of x when the lower rod (the one with length d) is vertical.

(Hint: At these points, P = π/2.) The answer is −0.74.

Determining the Speed at a Point

We now determine the speed of the point x. We need to use equation (1), and remember that everything will be a function of t, time. We again start from the original equations,

eq1 and eq2. We’ll form EQ again as before, but first express the idea that x, T , and P

are functions of t. Then, we can differentiate the equations with respect to t.

EQ := subs(x=x(t), P=P(t), T=T(t), {eq1, eq2});

DE := diff(EQ, t);

Now, we’ll let eq3 be equation (1), and add this to our list of equations, by substituting into DE. Then the unknown functions will be the derivatives (with respect to t) of x(t) and P (t).

eq3 := T(t) = a*cos(b*t);

DE := subs(eq3, DE);

DE := simplify(DE);

From the set of equations DE, solve for the derivatives of x(t) and P (t)

sol := solve(DE, {diff(x(t),t), diff(P(t),t)});

As before, sol is a list. One of the elements will be the solution for the derivative of x(t), the other will be the solution for the derivative of P (t). Set the variable dxdt to be the solution for the derivative of x(t). For example, you might have to type:

dxdt := sol[2]; Use sol[1] or sol[2], as appropriate

We want the right-hand side of the relation dxdt. To do this, we use the command

op(2, dxdt); This extracts the second operand (the right-hand side) from dxdt. This

expression is evaluated at t = π/(2b) since this is where T (t) = 0. Remember that we wanted to find the speed when T (t) = 0.

dxdt := op(2, dxdt);
dxdt0 := subs(t=Pi/(2*b), dxdt);
dxdt0 := simplify(dxdt0);

Check to see what happens if the op command is omitted in the above sequence.

What is the speed of x at T (t) = 0 if a = 0.5, b = 1.2, c = 0.8, and d = 0.7? The

answer is 0.48. What is the significance of the positive sign in this value? Why couldn’t it have been negative? (Hint: If you use t = 3π/(2b) for t (instead of π/(2b)), then answer is negative! At this value for t, T (t) = 0 as well.)